(%i1)
```
/* eq_R is the equation relating R(t) to t */
/* eq_x is the equation relating x(R) to R */
/* R_i is a function of R and x */

eq_R: R = 1+3*t*R**2;
eq_x: x+1/x = (4-R)/(R-1);
Ri: R*(1-x**i)*(1-x**(i+3))/(1-x**(i+1))/(1-x**(i+2));
```

(%o1) $R=3\,t\,R^2+1$

(%o2) $x+\frac{1}{x}=\frac{4-R}{R-1}$

(%o3) $\frac{\left(1-x^i\right)\left(1-x^{i+3}\right)R}{\left(1-x^{i+1}\right)\left(1-x^{i+2}\right)}$

(%i4)
```
/* solve eq_R for R(t), call R1,R2 the two solutions. Idem for x(R) */
/* R_sol is the function R(t) given by the first solution. Idem for x_sol */

[R1,R2]: solve(eq_R,R);
[x1,x2]: solve(eq_x,x);
R_sol : subst(R1,R);
x_sol : subst(x1,x);
```

(%o4) $[R=-\frac{\sqrt{1-12\,t}-1}{6\,t},R=\frac{\sqrt{1-12\,t}+1}{6\,t}]$

(%o5) $[x=-\frac{\sqrt{3}\sqrt{4-R^2+R-4}}{2\,R-2},x=\frac{\sqrt{3}\sqrt{4-R^2}-R+4}{2\,R-2}]$

(%o6) $-\frac{\sqrt{1-12\,t}-1}{6\,t}$

(%o7) $-\frac{\sqrt{3}\sqrt{4-R^2}+R-4}{2\,R-2}$

(%i8)
```
/* compute the "critical values" for t,R and x */
t_c: 1/12;
R_c: subst(t=t_c,R_sol);
x_c: subst(R=R_c,x_sol);
```

(%o8) $\frac{1}{12}$

(%o9) 2

(%o10) 1

(%i11)
```
/* In the following we use the variable dt=t_c-t instead of t, idem for R
and x */
/* DR: the Taylor expansion of dR(dt) up to order dt**(3/2) */
/* Dx: the Taylor expansion of dx(dR) up to order dR**3 */
/*DRi: the Taylor expansion of Ri(dx,dR) up to order dR**1 and dx**6 */

DR: taylor( R_c-subst(t=t_c-dt, R_sol), [dt,0,3/2] );
Dx: taylor( x_c-subst(R=R_c-dR, x_sol), [dR,0,3] );
DRi: taylor( subst([x=x_c-dx, R=R_c-dR], Ri), [dR,0,3], [dx,0,6] );
```

(%o11)/T/ $4\sqrt{3}\sqrt{dt}-24\,dt+48\sqrt{3}\,dt^{3/2}+\ldots$

(%o12)/T/ $\sqrt{3}\sqrt{dR}-\frac{3\,dR}{2}+\frac{7\sqrt{3}\,dR^{3/2}}{8}-\frac{3\,dR^2}{2}+\frac{37\sqrt{3}^3\,dR^{5/2}}{128}-\frac{3\,dR^3}{2}+\ldots$

(%o13)/T/
$$\frac{2i^2+6i}{i^2+3i+2}+\frac{(i^2+3i)\,dx^2}{3i^2+9i+6}+\frac{(i^2+3i)\,dx^3}{3i^2+9i+6}-\frac{(3i^4+18i^3-25i^2-156i)\,dx^4}{180i^2+540i+360}-\frac{(3i^4+18i^3+5i^2-66i)\,dx^5}{90i^2+270i+180}+\frac{(5i^6+45i^5-208i^4-1923i^3-1515i^2+4716i)\,dx^6}{7560i^2+22680i+15120}+\ldots+\Bigg(-\frac{i^2+3i}{i^2+3i+2}-\frac{(i^2+3i)\,dx^2}{6i^2+18i+12}-\frac{(i^2+3i)\,dx^3}{6i^2+18i+12}+\frac{(3i^4+18i^3-25i^2-156i)\,dx^4}{360i^2+1080i+720}+\frac{(3i^4+18i^3+5i^2-66i)\,dx^5}{180i^2+540i+360}-\frac{(5i^6+45i^5-208i^4-1923i^3-1515i^2+4716i)\,dx^6}{15120i^2+45360i+30240}+\ldots\Bigg)\,dR+$$

(%i14)
```
/* Now we substitute dR and dx by their expansion w.r.t. dt */
/* Dx_t: the Taylor expansion of dx(dt) up to order dt**(5/4) */
/*DRi_t: the Taylor expansion of Ri(dt) up to order dt**(3/2) */

Dx_t : subst(dR=DR,Dx);
DRi_t: subst([dR=DR,dx=Dx_t],DRi);
```

(%o14)/T/ $2\left(3^{1/4}\right)^3 dt^{1/4}-6\sqrt{3}\sqrt{dt}+15\,3^{1/4}\,dt^{3/4}-36\,dt+\frac{117\left(3^{1/4}\right)^3 dt^{5/4}}{4}+\ldots$

(%o15)/T/
$$\frac{2i^2+6i}{i^2+3i+2}-\frac{(36i^4+216i^3+300i^2-72i)\,dt}{5i^2+15i+10}+\frac{\left(120\sqrt{3}\,i^6+1080\sqrt{3}\,i^5+3576\sqrt{3}\,i^4+5256\sqrt{3}\,i^3+3120\sqrt{3}\,i^2+288\sqrt{3}\,i\right)dt^{3/2}}{35i^2+105i+70}+\ldots$$

(%i16)
```
/* extract C_i, the coefficient of dt**(3/2) in Ri(dt) */
/* compute C_i/C_1, which is the limit asked in Question 5 */

C_i: factor( coeff(DRi_t,dt**(3/2)) );
C_i/subst(i=1,C_i);
```

(%o16) $\dfrac{8\,3^{3/2}\,i\,(i+3)\,(5i^4+30i^3+59i^2+42i+4)}{35\,(i+1)\,(i+2)}$

(%o17) $\dfrac{3\,i\,(i+3)\,(5i^4+30i^3+59i^2+42i+4)}{280\,(i+1)\,(i+2)}$

Created with wxMaxima.